
Inteligencia Artificial 2 Curso 2000–2001

Tema 4: Ingenieŕıa del

Miguel A. Gutiérrez Naranjo

Francisco J. Mart́ın Mateos

Dpto. de Ciencias de la Computación e Inteligencia Artificial

Universidad de Sevilla

Introducción

x Contenido del tema
u Arquitectura de los SBC (sistemas basados en el

conocimiento)

u Construcción de SBC mediante metaintérpretes:

* Ejemplo de BC objeto

* Metaintérprete simple

* Metaintéprete con lenguaje ampliado

* Metaintérprete con profundidad acotada

u Herramientas para consulta al usuario

u Herramientas para explicación al usuario:

* Metaintérprete con Cómo

* Metaintérprete con Porqué

u Técnicas de depuración

u Bibliograf́ıa:

Poole–98 Cap. 5: “Knowledge engineering”

Arquitectura de los SBC

x Arquitectura de los SBC (Poole–98 p. 200)

Ejemplo de BC objeto

x El sistema eléctrico (Poole–98 p. 16)

Ejemplo de BC objeto

x Ejemplo de BC objeto: i electrica.pl

u Operadores

:- op(1100, xfx, <-).
:- op(1000, xfy, &).

u luz(?L) se verifica si L es una luz

luz(l1) <- verdad.
luz(l2) <- verdad.

u abajo(?I) se verifica si el interruptor I está hacia abajo

abajo(i1) <- verdad.

u arriba(?I) se verifica si el interruptor I está hacia ar-

riba

arriba(i2) <- verdad.
arriba(i3) <- verdad.

u esta bien(?X) se verifica si la luz (o cortacircuito) X

está bien.

esta_bien(l1) <- verdad.
esta_bien(l2) <- verdad.
esta_bien(cc1) <- verdad.
esta_bien(cc2) <- verdad.

Ejemplo de BC objeto

u conectado(?D1,?D2) se verifica si los dispositivos D1 y D2

está conectados (de forma que puede fluir la corriente

eléctrica de D2 a D1)

conectado(l1,c0) <- verdad.
conectado(c0,c1) <- arriba(i2).
conectado(c0,c2) <- abajo(i2).
conectado(c1,c3) <- arriba(i1).
conectado(c2,c3) <- abajo(i1).
conectado(l2,c4) <- verdad.
conectado(c4,c3) <- arriba(i3).
conectado(e1,c3) <- verdad.
conectado(c3,c5) <- esta_bien(cc1).
conectado(e2,c6) <- verdad.
conectado(c6,c5) <- esta_bien(cc2).
conectado(c5,entrada) <- verdad.

u tiene corriente(?D) se verifica si el dispositivo D tiene

corriente

tiene_corriente(D) <-
conectado(D,D1) &
tiene_corriente(D1).

tiene_corriente(entrada) <- verdad.

u esta encendida(?L) se verifica si la luz L está encendida

esta_encendida(L) <-
luz(L) &
esta_bien(L) &
tiene_corriente(L).

Metaintérprete simple

x Sesión

?- prueba(esta_encendida(X)).
X = l2 ;
No

x Metaintérprete simple

u prueba(+O) se verifica si el objetivo O se puede de-

mostrar a partir de la BC objeto

prueba(verdad).
prueba((A & B)) :-

prueba(A),
prueba(B).

prueba(A) :-
(A <- B),
prueba(B).

Metaintéprete ampliado

x Ampliación del lenguaje base:

u Disyunciones: A ; B

u Predicados predefinidos: is, <, ...

x Operadores

:- op(1100, xfx, <-).
:- op(1000, xfy, [&,;]).

x Ejemplo de BC ampliada

vecino(X,Y) <- Y is X-1 ; Y is X+1.

x Sesión

?- prueba(vecino(2,3)).
Yes
?- prueba(vecino(3,2)).
Yes

Metaintéprete ampliado

x Metaintéprete ampliado

u prueba(+O) se verifica si el objetivo O se puede de-

mostrar a partir de la BC objeto (que puede contener

disyunciones y predicados predefinidos)

prueba(verdad).
prueba((A & B)) :-

prueba(A),
prueba(B).

prueba((A ; B)) :-
prueba(A).

prueba((A ; B)) :-
prueba(B).

prueba(A) :-
predefinido(A),
A.

prueba(A) :-
(A <- B),
prueba(B).

u predefinido(+O) se verifica si O es un predicado pre-

definido

predefinido((X is Y)).
predefinido((X < Y)).

Metaintérprete con profundidad acotada

x Metaintérprete con profundidad acotada

u prueba pa(+O,+N) es verdad si el objetivo O se puede

demostrar con profundidad N como máximo

prueba_pa(verdad,_N).
prueba_pa((A & B),N) :-

prueba_pa(A,N),
prueba_pa(B,N).

prueba_pa(A,N) :-
N >= 0,
N1 is N-1,
(A <- B),
prueba_pa(B,N1).

x Ejemplo

numero(0) <- verdad.
numero(s(X)) <- numero(X).

lista(0,[]) <- verdad.
lista(s(X),[a|L]) <-

numero(s(X)) &
lista(X,L).

Metaintérprete con profundidad acotada

x Sesión

?- prueba_pa(numero(N), 3).
N = 0 ;
N = s(0) ;
N = s(s(0)) ;
N = s(s(s(0))) ;
No

?- prueba_pa(numero(s(s(0))),1).
No

?- prueba_pa(numero(s(s(0))),2).
Yes

?- prueba_pa(lista(N, L), 3).
N = 0 L = [] ;
N = s(0) L = [a] ;
N = s(s(0)) L = [a, a] ;
No

Metaintérprete con profundidad acotada

x Segundo ejemplo

u Programa

hermano(X,Y) <- hermano(Y,X).
hermano(b,a) <- verdad.

u Sesión

?- prueba_pa(hermano(a,X),1).
X = b ;
No

?- prueba_pa(hermano(X,Y),1).
X = a Y = b ;
X = b Y = a ;
No

?- prueba_pa(hermano(a,X),2).
X = b ;
No

?- prueba_pa(hermano(a,X),3).
X = b ;
X = b ;
No

Consulta al usuario

x Planteamiento del problema: Aportación del
conocimiento de los usuarios cuando:

u No conocen las interioridades del sistema

u No son expertos en el dominio

u No concocen qué información es importante

u No conocen la sintaxis del sistema

u Poseen información esencial del caso particular del

problema

x Funciones del sistema:
u Determinar qué información es importante

u Preguntar al usuario sobre dicha información

x Tipos de objetivos:

u No preguntable

u Preguntable no–preguntado

u Preguntado

Consulta al usuario

x Preguntas elementales:

u Son objetivos básicos (sin variables)

u Las respuestas son “śı” o “no”

u Se plantean si son importantes, preguntables y no pre-

guntadas

u El sistema almacena la respuesta

x Ejemplo: Modificación de i electrica.pl

preguntable(arriba(_)).
preguntable(abajo(_)).

x Sesión

?- prueba_p(esta_encendida(L)).
¿Es verdad arriba(i2)? (si/no)
|: si.
¿Es verdad arriba(i1)? (si/no)
|: no.
¿Es verdad abajo(i2)? (si/no)
|: no.
¿Es verdad arriba(i3)? (si/no)
|: si.

L = l2 ;
No

Consulta al usuario

?- listing(respuesta).
respuesta(arriba(i2), si).
respuesta(arriba(i1), no).
respuesta(abajo(i2), no).
respuesta(arriba(i3), si).
Yes

?- retractall(respuesta(_,_)).
Yes

?- prueba_p(esta_encendida(L)).
¿Es verdad arriba(i2)? (si/no)
|: si.
¿Es verdad arriba(i1)? (si/no)
|: si.
L = l1 ;

¿Es verdad abajo(i2)? (si/no)
|: no.
¿Es verdad arriba(i3)? (si/no)
|: si.
L = l2 ;
No

Consulta al usuario

x Metaintérprete con preguntas

u prueba p(+O) se verifica si el objetivo O se puede de-

mostrar a partir de la BC objeto y las respuestas del

usuario

prueba_p(verdad).
prueba_p((A & B)) :-

prueba_p(A),
prueba_p(B).

prueba_p(G) :-
preguntable(G),
respuesta(G,si).

prueba_p(G) :-
preguntable(G),
no_preguntado(G),
pregunta(G,Respuesta),
assert(respuesta(G,Respuesta)),
Respuesta=si.

prueba_p(A) :-
(A <- B),
prueba_p(B).

u respuesta(?O,?R) se verifica si la respuesta al objetivo

O es R. [Se añade dinámicamente a la base de datos]

:- dynamic respuesta/2.

IA2 2000–01 CcIa Ingenieŕıa del conocimiento y metaintérpretes 4.16



Consulta al usuario

u no preguntado(+O) es verdad si el objetivo O no se ha

preguntado

no_preguntado(O) :-
not(respuesta(O,_)).

u pregunta(+O, -Respuesta) pregunta O al usuario y éste

responde la Respuesta

pregunta(O,Respuesta) :-
escribe_lista([’¿Es verdad ’,O,’? (si/no)’]),
read(Respuesta).

u escribe lista(+L) escribe cada uno de los elementos

de la lista L

escribe_lista([]) :- nl.
escribe_lista([X|L]) :-

write(X),
escribe_lista(L).

Consulta al usuario

x Consulta sobre relaciones funcionales
u Ejemplo: edad(P,N)

x Forma de la consulta:
u Menús

u Texto de formato libre

x Preguntas generales

Pregunta ¿Preguntable? Respuesta
p(X) Sı́ p(f(Z))
p(f(c)) No
p(a) Sı́ Sı́
p(X) Sı́ No
P(X) No

Explicación

x Necesidad del sistema de justificar sus respues-
tas

x Uso en explicación y depuración

x Tipos de explicaciones:

u Preguntar Cómo se ha probado un objetivo

u Preguntar Porqué plantea una consulta

u Preguntar Porqué no se ha probado un objetivo

x Preguntas Cómo

u El usuario puede preguntar Cómo ha probado el ob-

jetivo q

u El sistema muestra la instancia de la regla usada

q :- p1,...,pn.

u El usuario puede preguntar Cómo i para obtener la

regla usada para probar pi

x El comando Cómo permite descender en el
árbol de prueba

Metaintérprete con Cómo

x Sesión

?- [meta_con_explicacion_como, i_electrica].
Yes

?- prueba_con_como(esta_encendida(L)).
esta_encendida(l2) :-

1: luz(l2)
2: esta_bien(l2)
3: tiene_corriente(l2)

|: 3.
tiene_corriente(l2) :-

1: conectado(l2, c4)
2: tiene_corriente(c4)

|: 2.
tiene_corriente(c4) :-

1: conectado(c4, c3)
2: tiene_corriente(c3)

|: 1.
conectado(c4, c3) :-

1: arriba(i3)
|: 1.
arriba(i3) es un hecho

L = l2 ;
No

Metaintérprete con Cómo

x Metaintérprete con Cómo

u prueba con como(+O) significa probar el objetivo O a

partir de la BC objeto y navegar por su árbol de

prueba mediante preguntas Cómo

prueba_con_como(O) :-
prueba_con_demostracion(O,A),
navega(A).

u navega(+A) significa que se está navegando en el árbol

A

navega(si(A,verdad)) :-
escribe_lista([A,’ es un hecho’]).

navega(si(A,B)) :-
B \== verdad,
escribe_lista([A,’ :-’]),
escribe_cuerpo(B,1,_),
read(Orden),
interpreta_orden(Orden,B).

u escribe lista(+L) escribe cada uno de los elementos

de la lista L

escribe_lista([]) :- nl.
escribe_lista([X|L]) :-

write(X),
escribe_lista(L).

Metaintérprete con Cómo

u escribe cuerpo(+B,+N1,?N2) es verdad si B es un cuerpo

que se va a escribir, N1 es el número de átomos antes

de la llamada a B y N2 es el número de átomos después

de la llamada a B

escribe_cuerpo(verdad,N,N).
escribe_cuerpo((A & B),N1,N3) :-

escribe_cuerpo(A,N1,N2),
escribe_cuerpo(B,N2,N3).

escribe_cuerpo(si(H,_),N,N1) :-
escribe_lista([’ ’,N,’: ’,H]),
N1 is N+1.

u interpreta orden(+Orden,+B) interpreta la Orden sobre

el cuerpo B

interpreta_orden(N,B) :-
integer(N),
nth(B,N,E),
navega(E).

u nth(+E,+N,?A) es verdad si A es el N–ésimo elemento de

la estructura E

nth(A,1,A) :-
not((A = (_,_))).

nth((A&_),1,A).
nth((_&B),N,E) :-

N>1,
N1 is N-1,
nth(B,N1,E).

Metaintérprete con Porqué

x Ejemplo: Modificación de i electrica.pl

preguntable(arriba(_)).
preguntable(abajo(_)).

x Sesión

?- [meta_con_explicacion_porque, i_electrica_con_preguntas].
Yes

?- prueba_con_porque(esta_encendida(L)).
¿Es verdad arriba(i2)? (si/no/porque)
|: porque.
Se usa en:

conectado(c0, c1) <-
arriba(i2).

¿Es verdad arriba(i2)? (si/no/porque)
|: porque.
Se usa en:

tiene_corriente(c0) <-
conectado(c0, c1) &
tiene_corriente(c1).

¿Es verdad arriba(i2)? (si/no/porque)
|: porque.
Se usa en:

tiene_corriente(l1) <-
conectado(l1, c0) &
tiene_corriente(c0).

Metaintérprete con Porqué

¿Es verdad arriba(i2)? (si/no/porque)
|: porque.
Se usa en:

esta_encendida(l1) <-
luz(l1) &
esta_bien(l1) &
tiene_corriente(l1).

¿Es verdad arriba(i2)? (si/no/porque)
|: porque.
Porque esa fue su pregunta!
¿Es verdad arriba(i2)? (si/no/porque)
|: si.
¿Es verdad arriba(i1)? (si/no/porque)
|: porque.
Se usa en:

conectado(c1, c3) <-
arriba(i1).

¿Es verdad arriba(i1)? (si/no/porque)
|: no.
¿Es verdad abajo(i2)? (si/no/porque)
|: no.
¿Es verdad arriba(i3)? (si/no/porque)
|: porque.
Se usa en:

conectado(c4, c3) <-
arriba(i3).

¿Es verdad arriba(i3)? (si/no/porque)
|: si.

L = l2 ;
No

té

rp
re

te
s

4.
28



M
e
ta

in
té

Metaintérprete con Porqué

x Metaintérprete con Porqué

u prueba con porque(+O) significa probar el objetivo O,

con las respuestas del usuario, permitiéndole pregun-

tar Porqué se le plantean preguntas

prueba_con_porque(O) :-
prueba_con_porque_aux(O,[]).

u prueba con porque aux(+O,+Antecesores) se verifica si O

es probable con la lista de Antecesores

[Pregunta al usuario y le permite preguntar Porqué ]

prueba_con_porque_aux(verdad,_).
prueba_con_porque_aux((A & B), Antecesores) :-

prueba_con_porque_aux(A,Antecesores),
prueba_con_porque_aux(B,Antecesores).

prueba_con_porque_aux(O,_) :-
preguntable(O),
respuesta(O,si).

prueba_con_porque_aux(O,Antecesores) :-
preguntable(O),
no_preguntado(O),
pregunta(O,Respuesta,Antecesores),
assert(respuesta(O,Respuesta)),
Respuesta=si.

prueba_con_porque_aux(A,Antecesores) :-
(A <- B),
prueba_con_porque_aux(B,[(A <- B)|Antecesores]).

IA2 2000–01 CcIa Ingenieŕıa del conocimiento y metaintérpretes 4.31



Metaintérprete con Porqué

u pregunta(+O,-Respuesta,+Antecesores) pregunta al

usuario, en el contexto dado por los Antecesores, la

cuestión O y éste responde la Respuesta

pregunta(O,Respuesta,Antecesores) :-
escribe_lista([’¿Es verdad ’,O,’? (si/no/porque)’]),
read(Replica),
interpreta(O,Respuesta,Replica,Antecesores).

u interpreta(+O,?Respuesta,+Replica,+Antecesores)

interpreta(_,Replica,Replica,_) :-
Replica \== porque.

interpreta(O,Respuesta,porque,[Regla|Reglas]):-
write(’Se usa en:’), nl,
escribe_regla(Regla),
pregunta(O,Respuesta,Reglas).

interpreta(O,Respuesta,porque,[]):-
write(’Porque esa fue su pregunta!’), nl,
pregunta(O,Respuesta,[]).

u escribe regla(Regla)

escribe_regla((A <- B)) :-
escribe_lista([’ ’,A, ’ <- ’]),
escribe_cuerpo(B).

IA2 2000–01 CcIa Ingenieŕıa del conocimiento y metaintérpretes 4.32



Metaintérprete con Porqué

u escribe cuerpo(C)

escribe_cuerpo((A & B)) :-
escribe_lista([’ ’,A, ’ &’]),
escribe_cuerpo(B).

escribe_cuerpo(A) :-
not(A = (_B & _C)),
escribe_lista([’ ’,A,’.’]).

x Utilidad de saber porqué el sistema plantea una
pregunta:

u Aumenta la confianza del usuario

u Ayuda al ingeniero del conocimiento en la opti-

mización de las preguntas realizadas por el sistema

u Una pregunta irrelevante puede manifestar la existen-

cia de problemas en el sistema

u El usuario puede usar el sistema para aprender

Depuración de bases de conocimiento

x Tipos de errores no sintácticos

u Respuestas incorrectas

u Respuestas perdidas

u Bucle infinito

u Preguntas irrelevantes

x Depuración de respuestas incorrectas

u Una respuesta incorrecta es una respuesta computada

que es falsa en la interpretación deseada

u Una respuesta incorrecta implica la existencia de una

cláusula de la BC que es falsa en la interpretación

deseada (o que el sistema de razonamiento es inade-

cuado)

u Si q es falsa en la interpretación deseada, existe una

demostración de q usando q :- p1,...,pn. Entonces

1. alguna pi es falsa (y se depura), o

2. todas las pi son verdaderas (y la cláusula es falsa)

Depuración de respuestas incorrectas
x Ejemplo

u BC errónea: en i electrica.pl cambiar la cláusula

conectado(c1,c3) <- arriba(i1).

por la cláusula errónea

conectado(c1,c3) <- arriba(i3).

u Depuración de respuesta errónea con el

metaintérprete con Cómo

?- [i_electrica_con_error_1, meta_con_explicacion_como].
Yes

?- prueba_con_como(esta_encendida(L)).
esta_encendida(l1) :-

1: luz(l1)
2: esta_bien(l1)
3: tiene_corriente(l1)

|: 3.
tiene_corriente(l1) :-

1: conectado(l1, c0)
2: tiene_corriente(c0)

|: 2.
tiene_corriente(c0) :-

1: conectado(c0, c1)
2: tiene_corriente(c1)

|: 2.
tiene_corriente(c1) :-

1: conectado(c1, c3)
2: tiene_corriente(c3)

|: 1.
conectado(c1, c3) :-

1: arriba(i3)

